

NOTICES

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever, and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

The Government has the right to reproduce, use, and distribute this report for governmental purposes in accordance with the contract under which the report was produced. To protect the proprietary interests of the contractor and to avoid jeopardy of its obligations to the Government, the report may not be released for non-governmental use such as might constitute general publication without the express prior consent of The Ohio State University Research Foundation.

Qualified requesters may obtain copies of this report from the Defense Documentation Center, Cameron Station, Alexandria, Virginia. Department of Defense contractors must be established for DDC services, or have their "need-to-know" certified by the cognizant military agency of their project or contract.

REPORT 1691-16

REPORT
by
THE OHIO STATE UNIVERSITY RESEARCH FOUNDATION
COLUMBUS, OHIO 43212

Sponsor	National Aeronautics and Space Administration Office of Grants and Research Contracts Washington, D. C.
Grant Number	NsG-448
Investigation of	Spacecraft Antenna Problems
Subject of Report	Higher-Order Diffraction Concept Applied to Parallel-Plate Waveguide Patterns
Submitted by	R. C. Rudduck and J. S. Yu Antenna Laboratory Department of Electrical Engineering
Date	15 October 1965

ABSTRACT

A method for computing the radiation patterns of parallel-plate waveguide apertures is presented. This method, which is based on edge diffraction theory, employs the Higher-Order Diffraction Concept. This concept permits the summation of all orders of diffraction in closed form. Calculated waveguide patterns are compared with those obtained by using the first three orders of diffraction.

This method is also applicable to other types of diffracting structures. Computations obtained by this method are, in general, quite accurate. However, the assumption is made that multiple diffractions are the same as those of isotropic line sources; this approximation leads to inaccurate results in some cases.

TABLE OF CONTENTS

	Page
I. INTRODUCTION	1
II. ANALYSIS OF WAVEGUIDE PATTERNS	1
III. COMPUTATIONS	10
IV. DISCUSSION	19
V. CONCLUSIONS	20
APPENDIX I	21
APPENDIX II.	26
REFERENCES	28

HIGHER-ORDER DIFFRACTION CONCEPT APPLIED TO PARALLEL-PLATE WAVEGUIDE PATTERNS

I. INTRODUCTION

Edge-diffraction theory has been applied to analyze diffraction by antenna apertures[1]. In this approach a singly diffracted wave emanates from each edge that is illuminated by the primary source. The singly diffracted waves again diffract from the various edges producing doubly diffracted waves. The process continues to higher and higher orders of diffraction.

In general multiple diffraction effects are quite significant. Previously, analyses were performed by computing the lowest orders of diffraction. Thus calculation of diffraction of a certain order was an iterative process in which all lower orders of diffraction were calculated. The iterative process is useful because the magnitude of diffraction diminishes rapidly with increasing order. However, it was discovered that all orders of diffraction can be included in calculations in a straightforward manner. The object of this report is to describe the concept of including all orders of diffraction which is denoted the "Higher-Order Diffraction Concept". This concept was subsequently found to be an application of the Self-Consistent Procedure which is used in scattering theory. The Higher-Order Diffraction Concept is illustrated by its application to the analysis of parallel-plate waveguide radiation patterns. The analysis of parallel-plate guide patterns by the iterative process is given in Refs. 2 and 3, in which the first three orders of diffraction are included in computations. The Higher-Order Diffraction Concept simplifies the formulation of solutions to diffraction problems and increases the accuracy of computations.

II. ANALYSIS OF WAVEGUIDE PATTERNS

The aperture of a parallel-plate waveguide is shown in Fig. 1. The guide angle, θ_g , between the edges of the waveguide walls may be adjusted for various mounting configurations, as may the wedge angles, $WA1$ and $WA2$. Only the TEM waveguide mode is considered in this report. In the TEM mode an incident plane wave propagates parallel to the axis of the guide with polarization perpendicular to the guide walls as shown in Fig. 1.




Fig. 1. TEM mode in a general parallel-plate waveguide.

The diffraction from the guide aperture is treated by superposing the diffracted waves from each of the wedges. The diffraction from each wedge is expressed in terms of the wedge diffraction function v_B [1, 2, 3, 4]. The computer subroutine used to calculate v_B is given in Appendix I. The incident plane wave on edge 1 gives rise to a diffracted wave

$$(1) \quad H_{D1}(r, \theta) = v_B(r, \pi + \theta),$$

where r is the distance from the edge and θ is defined in Fig. 1. H_{D1} is the value of the perpendicular component of the field (magnetic field in this case) for a unit-amplitude incident field. At large distances from the edge the propagation factor

$$\frac{e^{-j\left(kr + \frac{\pi}{4}\right)}}{\sqrt{2\pi kr}}$$

is separable from v_B . Thus the singly diffracted rays may be written as

$$(2) \quad R_{D1}(\theta) = \frac{1}{n_1} \sin \frac{\pi}{n_1} \left(\frac{1}{\cos \frac{\pi}{n_1} - \cos \frac{\pi + \theta}{n_1}} \right) ,$$

where the propagation factor is suppressed because only angular variations are of interest here. The constants n (non-integers in general) specify the wedge angles; i. e., $WA = (2-n)\pi$.

Similarly, the single diffracted wave from edge 2 is obtained as

$$(3) \quad R_{D2}(\theta) = \frac{1}{n_2} \sin \frac{\pi}{n_2} e^{-jka \cot \theta_g} \left(\frac{1}{\cos \frac{\pi}{n_2} - \cos \frac{\pi - \theta}{n_2}} \right) ,$$

where the factor $e^{-jka \cot \theta_g}$ represents the phase of the incident wave at edge 2 with respect to edge 1.

The direction $\theta = 0$ corresponds to the direction of the geometrical optics rays. Although the individual rays R_{D1} and R_{D2} are without limit at $\theta = 0$, the limit of $(R_{D1} + R_{D2})$ as $\theta \rightarrow 0$ expresses the effect of the geometrical optics rays.

Some of the singly diffracted rays from edge 1 may be reflected from wedge 2 as seen in Fig. 1. The reflected rays are given by

$$(4) \quad R_{RFL}(\theta) = R_{D1}(-\theta), \quad \theta_g < \theta < \frac{\pi}{2} .$$

Equations (2) and (3) express the diffraction of the incident plane wave by wedges 1 and 2, respectively. Equation (4) (the reflected wave) expresses an interaction between the two wedges. Other interactions which arise are the multiple diffractions between the edges. The singly diffracted wave from edge 1 again diffracts from edge 2, resulting in a doubly diffracted wave as shown in Fig. 2. The reflected wave and the singly diffracted wave from edge 2 illuminate edge 1 giving rise to two additional doubly diffracted waves. The doubly diffracted waves in turn




Fig. 2. Doubly-diffracted rays.

produce triply diffracted waves and the process continues indefinitely to higher and higher orders of diffraction. The magnitude of each order of diffraction diminishes with increasing order. In general, double diffraction must be included for reasonable results and triple- and higher-order diffractions are often necessary. The Higher-Order Diffraction Concept permits the inclusion of all orders of diffraction in a simple manner.

The doubly diffracted waves can be treated as the diffraction of cylindrical waves because the singly diffracted waves behave as cylindrical waves radiating from the edges of the waveguide aperture. Thus the doubly diffracted wave from edge 2 is given by

$$(5) \quad R_{DD2}(\theta) = R_{D1G} [v_B(h, \pi - \theta - \theta_g) + v_B(h, \pi - \theta + \theta_g)] ,$$

where the illuminating cylindrical wave intensity from edge 1 is given by

$$(6) \quad R_{D1G} = R_{D1}(-\theta_g).$$

However, the Higher-Order Diffraction Concept permits the determination of the total higher-order diffraction from edge 2 by using the illumination of the total cylindrical wave from edge 1. The total illumination from edge 1 may be expressed as

$$(7) \quad R_{1G} = R_1(-\theta g) ,$$

where $R_1(\theta)$ is the total diffracted wave from edge 1 as shown in Fig. 3. Consequently, the total higher-order diffraction (i. e., second-order and higher) from edge 2 is given by

$$(8) \quad R_{HD2}(\theta) = R_{1G}[v_B(h, \pi - \theta - \theta g) + v_B(h, \pi - \theta + \theta g)] .$$

Thus the total diffracted wave from edge 2 is given by

$$(9) \quad R_2(\theta) = R_{D2}(\theta) + R_{HD2}(\theta) .$$




Fig. 3. Total diffracted rays.

It should be noted that R_{1G} is unknown at this point in the formulation, whereas R_{D1G} is known. The key feature of the Higher-Order Diffraction Concept is that it allows R_{1G} and other unknown illuminating rays to be determined from a set of simultaneous linear equations. This aspect of the Higher-Order Diffraction Concept will be developed at a subsequent point in the analysis.

Now consider the two doubly diffracted waves from edge 1. The doubly diffracted wave caused by the singly diffracted ray,

$$(10) \quad R_{D2G} = R_{D2}(\pi - \theta_g),$$

is given by

$$(11) \quad R_{DD1}(\theta) = R_{D2G} [v_B(h, \theta + \theta_g) + v_B(h, 2\pi + \theta - \theta_g)] .$$

The double diffraction by the singly diffracted ray,

$$(12) \quad R_{D1P} = R_{D1}(-\pi/2),$$

is given by

$$(13) \quad R_{D1RD1}(\theta) = R_{D1P} \left[v_B \left(2a, \frac{\pi}{2} + \theta \right) + v_B \left(2a, \frac{3\pi}{2} + \theta \right) \right] .$$

The total illumination of edge 1 from edge 2 is given by

$$(14) \quad R_{2G} = R_2 (\pi - \theta_g)$$

and the total illumination of edge 1 from the reflection of the diffracted wave from edge 1 is given by

$$(15) \quad R_{1P} = R_1(-\pi/2) .$$

Thus the total higher-order diffraction from edge 1 is given by

$$(16) \quad R_{HDI}(\theta) = R_{2G}[v_B(h, \theta + \theta_g) + v_B(h, 2\pi + \theta - \theta_g)] \\ + R_{1P}[v_B\left(2a, \frac{\pi}{2} + \theta\right) + v_B\left(2a, \frac{3\pi}{2} + \theta\right)] .$$

Consequently the total diffraction from edge 1 is given by

$$(17) \quad R_1(\theta) = R_{D1}(\theta) + R_{HDI}(\theta) .$$

The principal aspect of the Higher-Order Diffraction Concept is that it permits the unknown illuminating rays R_{1G} , R_{2G} , and R_{1P} to be determined. Three simultaneous linear equations can be formulated in terms of the three unknown rays by use of Eqs. (9) and (17). Thus

$$(18) \quad R_{1G} = R_{D1G} + R_{2G} V_{2G}(-\theta_g) + R_{1P} V_{1P}(-\theta_g), \\ R_{1P} = R_{D1P} + R_{2G} V_{2G}(-\pi/2) + R_{1P} V_{1P}(-\pi/2) , \text{ and} \\ R_{2G} = R_{D2G} + R_{1G} V_{1G}(\pi - \theta_g),$$

where the quantities V_{1G} , V_{1P} , and V_{2G} are the unit wave diffractions used in Eqs. (8) and (16). The simultaneous equations of (18) form a set of three complex equations in three complex unknowns. This set of equations is equivalent to six real equations in six real unknowns.

Upon determination of the unknown rays the total diffracted wave from the aperture may be expressed as the superposition of the total diffracted waves from edges 1 and 2 plus the total reflected wave to yield

$$(19) \quad R_T(\theta) = R_1(\theta) + R_2(\theta) e^{jkh \cos(\theta + \theta_g)} \\ + R_1(-\theta) e^{-j2ka \sin \theta_g \sin \theta} .$$

The exponential factors refer each term to a common phase reference at edge 1. Each term in Eq. (19) contributes to the radiation pattern only in certain regions as follows:

$$\begin{aligned}
 (20) \quad R_1(\theta): & \quad -\theta_g < \theta < \pi - WA1, \\
 R_2(\theta): & \quad \pi + WA2 < \theta < \pi - \theta_g, \text{ and} \\
 R_1(-\theta): & \quad +\theta_g < \theta < \pi/2.
 \end{aligned}$$

Normally truncated waveguides ($\theta_g = 90^\circ$) must be treated somewhat differently because no reflected rays contribute to the radiation pattern and

$$(21) \quad R_{1P} = 0 \quad \text{for } \theta_g = 90^\circ.$$

In this case there are only two unknown rays, R_{1G} and R_{2G} ; and the simultaneous equations reduce to

$$\begin{aligned}
 (22) \quad \theta_g = 90^\circ: \\
 R_{1G} &= R_{D1G} + R_{2G} V_{2G}(-\pi/2) \quad \text{and} \\
 R_{2G} &= R_{D2G} + R_{1G} V_{1G}(\pi/2).
 \end{aligned}$$

The total pattern for the $\theta_g = 90^\circ$ case is given by

$$\begin{aligned}
 (23) \quad \theta_g = 90^\circ: \\
 R_T(\theta) &= R_1(\theta) + R_2(\theta) e^{-jka \sin \theta},
 \end{aligned}$$

where the appropriate regions for the respective terms are

$$\begin{aligned}
 (24) \quad R_1(\theta): & \quad -\pi/2 < \theta < \pi \quad \text{and} \\
 R_2(\theta): & \quad -\pi < \theta < \pi/2.
 \end{aligned}$$

The solution to the simultaneous equation of Eq. (22) is readily obtained as

$$(25) \quad R_{1G} = \frac{R_{D1G} + R_{D2G} V_{2G} \left(-\frac{\pi}{2}\right)}{1 - V_{1G} \left(\frac{\pi}{2}\right) V_{2G} \left(-\frac{\pi}{2}\right)}$$

and

$$(26) \quad R_{2G} = \frac{R_{D2G} + R_{D1G} V_{1G} \left(\frac{\pi}{2}\right)}{1 - V_{2G} \left(-\frac{\pi}{2}\right) V_{1G} \left(\frac{\pi}{2}\right)}$$

The values of the illuminating rays of Eqs. (25) and (26) can also be obtained by use of the summation of a geometric series. This technique was described by Burke and Keller[5] for the thick rectangular edge. By the process of iteration the illuminating ray from edge 1 can be expressed as

$$(27) \quad \begin{aligned} R_{1G} = & R_{D1G} + R_{D1G} V_{1G}(\pi/2) V_{2G}(-\pi/2) \\ & + R_{D1G} [V_{1G}(\pi/2) V_{2G}(-\pi/2)]^2 + \dots \\ & + R_{D2G} V_{2G}(-\pi/2) + R_{D2G} [V_{2G}(-\pi/2)]^2 V_{1G}(\pi/2) \\ & + R_{D2G} V_{2G}(-\pi/2) [V_{2G}(-\pi/2) V_{1G}(\pi/2)]^2 + \dots \end{aligned}$$

The sum of the geometrical progression which occurs in Eq. (27) can be employed to obtain Eq. (25). Equation (26) may be derived in the same manner. For cases in which there are more than two illuminating rays it is necessary to use the Higher-Order Diffraction Concept to evaluate each illuminating ray.

The process of expressing the interactions between two edges in terms of a geometrical progression is known in scattering theory as the Successive Scattering Procedure[6]. Moreover, the Higher-Order Diffraction Concept was found to be an application of the Self-Consistent Procedure used in scattering theory[6].

III. COMPUTATIONS

In order to compute the radiation pattern from Eq. (19) the simultaneous equations of Eq. (18) must be solved. Since, computations are made on a digital computer a numerical method for solving simultaneous linear equations which are in complex form is desirable. The Crout method[7] for solving simultaneous equations is quite suitable for this purpose. The computer subroutine of Richmond[8], as presented in Appendix II, employs the Crout method and is used to solve the simultaneous equations. The accuracy of results of the subroutine was tested by substituting the solved values for the unknowns into the equations. The accuracy of the solutions to the simultaneous equations was found to be excellent in all cases. The computer subroutine for computing the diffraction function $V_B(r, \phi)$ is given in Appendix I.

The wide variety of pattern shapes available from parallel-plate waveguides is illustrated by Figs. 4 - 10. The patterns of Figs. 4 and 5 were also calculated by the iterative method of Refs. 2 and 3, in which the first three orders of diffraction were included. It is evident that the iterative method gives very good results for these cases.

The iterative method gives increasingly accurate results for larger and larger guide widths. The accuracy decreases accordingly for small guide width. In order to compare the iterative method with the Higher-Order Diffraction method, patterns were computed by both methods for very small guide widths. In Fig. 6 the patterns of a $\theta_g = 90^\circ$ guide with a guide width of $a = 0.2\lambda$ are shown. The results of the iterative method agree very closely except near the plane of the aperture. The discontinuity in the pattern computed by the iterative method results from not including the higher-order diffractions. Other comparisons of the two methods are given in Figs. 7 and 8 for a $\theta_g = 60^\circ$ guide. As seen from Fig. 7 the results of the two methods agree closely for a guide width of $a = 0.3\lambda$. For a guide width of $a = 0.2\lambda$ the difference is more significant, as shown in Fig. 8.

Some general aspects of computing patterns by edge diffraction theory can be seen from observation of the figures. In Fig. 9 a discontinuity occurs at $\theta = 180^\circ$ as a result of the semi-infinite guide which extends in the 180° direction. This type of discontinuity does not occur in Fig. 4 for the normally truncated guide because of the symmetry of the guide and its patterns. In Fig. 10 it is seen that the boundary conditions for the $\theta_g = 90^\circ$ guide flush mounted to a ground plane are not satisfied exactly because the slope of the pattern is not zero in the directions of the ground plane surface. The reason for the non-zero slope is the




Fig. 4. Pattern for $\theta_g = 90^\circ$, $WA_1 = WA_2 = 0$, $a/\lambda = 0.625$.




Fig. 5. Pattern for $\theta_g = 30^\circ$, $WA1 = 30^\circ$, $WA2 = 150^\circ$, $a/\lambda = 0.423$.




Fig. 6. Pattern for $\theta_g = 90^\circ$, $WA_1 = WA_2 = 0$, $a/\lambda = 0.200$.




Fig. 7. Pattern for $\theta_g = 60^\circ$, $WA_1 = WA_2 = 0$, $a/\lambda = 0.300$.




Fig. 8. Pattern for $\theta_g = 60^\circ$, $WA_1 = WA_2 = 0$, $a/\lambda = 0.200$.




Fig. 9. Pattern for $\theta_g = 30^\circ$, $WA_1 = WA_2 = 0$, $a/\lambda = 0.625$.




Fig. 10. Pattern for $\theta_g = 90^\circ$, $WA1 = WA2 = 90^\circ$, $a/\lambda = 0.625$.

assumption used in the application of edge diffraction theory that each diffracted wave is the same as that for uniform incident wave. In the guide of Fig. 10 the total wave from edge 1 illuminating edge 2 is approximated as the uniform wave of an isotropic line source. However, the wave from edge 1 is not uniform, thus resulting in slight inaccuracy near the ground plane. The same approximation applies to Fig. 5 but the slope of the incident wave is not noticeable.

The edge-diffraction method for computing parallel-plate guide patterns generally gives accurate results. This is indicated by the comparison with measured patterns given in Refs. 2 and 3. The edge diffraction method fails in some cases because of the basic assumption that multiple diffractions are the same as the diffraction of an isotropic line source.

The edge diffraction method will now be compared with conventional methods of pattern calculation. The Kirchoff method assumes that the total field in the plane of the aperture is equal to the incident field in the aperture and is zero on the remaining surface of the aperture, and thus gives the pattern as

$$(28) \quad |E_K| = \frac{\sin\left(\frac{1}{2} ka \sin \theta\right)}{\left(\frac{1}{2} ka \sin \theta\right)}, \quad -90^\circ < \theta < 90^\circ.$$

The approximation of the Kirchoff method is most accurate for the ground plane case, i. e., $WA1 = \theta_g$ and $WA2 = \pi - \theta_g$. The pattern for the guide geometry of Fig. 10 is also calculated by Eq. (28). As is evident, the Kirchoff method is more accurate in this case because of the limitation of the edge-diffraction method which was previously discussed. However, the Kirchoff method is limited in the general case in that it is impractical to calculate fields behind the plane of the aperture.

The Wiener-Hopf technique may be applied to the case for which $\theta_g = 90^\circ$ and $WA1 = WA2 = 0$ and yields the exact solution for this case. The Wiener-Hopf pattern is given by [9]

$$(29) \quad |E_W| = \frac{\sin\left(\frac{1}{2} ka \sin \theta\right)}{\left(\frac{1}{2} ka \sin \theta\right)} \exp\left(\frac{1}{4} ka \cos \theta\right).$$

The pattern for the guide geometry of Fig. 4 is also calculated by Eq. (29). It is seen from Fig. 4 that the edge diffraction method is quite accurate for this guide with a width of only 0.338λ . The Wiener-Hopf technique is practically limited to the special case of the thin-walled, normally truncated guide.

IV. DISCUSSION

The formulation of a problem with the Higher-Order Diffraction Concept is as simple as that necessary to include double diffractions. The use of the Higher-Order Diffraction Concept is somewhat more lengthy than use of the first and second orders of diffractions in that a set of simultaneous linear equations must be solved. However, this is relatively simple with a digital computer which is usually necessary to calculate diffraction solutions. Thus the Higher-Order Diffraction concept provides the high accuracy of including many orders of multiple diffractions for an insignificant increase in complexity of solving the problem.

The Higher-Order Diffraction Concept can be applied to any two-dimensional diffraction problem treated by edge-diffraction theory. The analysis of any structure formed by superposition of wedges may be treated in the same manner as the parallel-plate waveguides are treated in this report. The total diffraction is composed of the singly diffracted waves caused by the primary illumination, and the total higher-order diffracted waves from each edge caused by the total illuminating rays from all visible edges and visible images of edges. The number of simultaneous equations is equal to the total number of unknown illuminating rays. The computer subroutine of Appendix II is applicable to any number of simultaneous equations.

The edge diffraction technique used in conjunction with the Higher-Order Diffraction Concept does not give exact formulations of diffraction solutions even though the results are generally quite accurate. The limitation results from the fact that multiple diffractions are assumed to be equivalent to diffraction by isotropic line sources. Although this assumption is generally quite good, it fails to be accurate in some cases. One case occurs in which the singly diffracted wave resulting from a plane wave illuminates another edge. If the other edge is too near the shadow boundary of the illuminating singly diffracted wave the assumption of uniform illumination is inaccurate. The reason is that the diffracted wave resulting from a plane wave has steep slopes near its shadow boundary. Another case in which the assumption of uniform illumination fails is for very small dimensions between edges where multiple diffractions

are computed. The assumption is generally good for dimensions larger than a small fraction of a wavelength. Thus limitations of the edge diffraction technique result from the assumption that diffraction is a local edge effect and do not result from the Higher-Order Diffraction Concept.

V. CONCLUSIONS

The concept of "Higher-Order Diffraction" is demonstrated by its application to the TEM radiation patterns of parallel-plate waveguides. This concept permits all orders of diffraction to be expressed in closed form. Thus the computation of diffraction by structures composed of wedges may generally be computed accurately and simply.

The limitation on the accuracy of the solution obtained by edge diffraction theory results from the assumption that multiple diffractions are the same as the diffraction by isotropic line sources. This assumption may not be valid in certain cases; e. g., when one edge is located near the shadow boundary of a singly diffracted wave.

APPENDIX I

The Scatran computer subroutine for the calculation of the diffraction function $V_B(r, \phi)$ is presented in this Appendix. The function V_B is calculated by either the Fresnel integral formulation of Pauli or the cylindrical wave formulation[1]. The Fresnel integral is computed by a highly accurate technique developed by Fleckner[10]. The cylindrical wave formulation was programmed by M. L. Tripp.

```

SUBROUTINE (RVB,UVB)=VB.(FK,ARG,ANGLE,FN) -
PROVIDED (ANGLE.LE.180.0*FN), TRANSFER TO (V1)-
ANGLE=ANGLE-360.*FN-
V1 R=FK*ARG/6.28318530-
PHI=ANGLE*3.14159265/180.-
PROVIDED(FN.NE.2.0.AND.R.L.1),TRANSFER TO (VDD)-
PRECISION (2,DSIN.,DCOS.,CONST,DPFN,DPANGR) -
PROVIDED (ANGLE.E.180..OR.ANGLE.E.-180.), TRANSFER TO (ALTER
) -
DPFN = FN -
CONST = 3.1415926536/DPFN -
ANGLER = ANGLE*3.14159265/180. -
DPANGR = ANGLE*3.1415926536/180. -
COSANG = COS.(ANGLER) -
X = FK*ARG*(1.+COSANG) -
CALL FUNCTION (C)=RFRESN.(X) -
CALL FUNCTION (S)=UFRESN.(X) -
RCOEFF = COS.(FK*ARG*COSANG) -
UCOEFF = SIN.(FK*ARG*COSANG) -
RREST = .5-C/2.-S/2. -
UREST = S/2.-C/2. -
COEF=(1./DPFN)*(DSIN.(CONST))*2.*.ABS.(DCOS.(DPANGR/2.))/
(DCOS.(CONST)-DCOS.(DPANGR/DPFN)) -
RVB=COEF*(RCOEFF*RREST-UCOEFF*UREST) -
UVB=COEF*(RCOEFF*UREST+UCOEFF*RREST) -
NORMAL EXIT -
ALTER RVB = .5*COS.(FK*ARG) -
UVB = -.5*SIN.(FK*ARG) -
NORMAL EXIT -
C VD SUBROUTINE-
VDD TRANSFER(VDD2)PROVIDED(R.E.PR.AND.FN.E.PFN)-
DIMENSION(Q(36))-
COMPLEX(CVD,COMZ,Y,POWR)-
LITERALS(PIOH,1.57079633,PI,3.14159265)-
LITERALS(PI2,6.28318531,COMZ,0.,1.0.)-
PR=R-
PFN=FN-
DOTHROUGH(VDD1),M=0.1,M/FN,LE.15.-
CALLSUBROUTINE(Q(Y))=BESSEL.(M/FN,PI2*R)-
VDD1 CONTINUE-
VDD2 CONTINUE-
CP=COS.(PHI)-
CVD=COMZ-
DOTHROUGH(VDD3),M=1.1,M/FN,LE.15.-
TEMP=PIOH*M/FN-
POWR=COS.(TEMP)+.I.SIN.(TEMP)-
CVD=CVD+POWR*Q(M)*COS.(M*PHI/FN)-
VDD3 CONTINUE-
CALLSUBROUTINE(Y)=CEXP.(0.+.I.PI2*R*CP)-
PROVIDED(.ABS.ANGLE.GE.180.),Y=COMZ-
CVD=2.0/FN*CVD-Y-
CVD=CVD+Q(0)/FN-
RVB=.REAL.CVD-
UVB=.IMAG.CVD-
NORMALEXIT-
ENDD ENDSUBPROGRAM-
SUBROUTINE (BESL)=BESSEL.(P,X) -
-
C -
C BESL IS THE BESSEL FUNCTION OF THE FIRST KIND, OF ORDER P,
EVALUATED AT AN ARGUMENT OF X. -
C -
PRECISION (2,DPF,DPX,DPK,DPTMP,DPBESL,DPCONS) -
PROVIDED (X.GE.0.), TRANSFER TO (BEGIN) -
BESL = 0. -

```

```

NORMAL EXIT -
BEGIN  TEMP = GAMMA.(P+1.) -
      DPTEMP = TEMP -
      DPTEMP = 1./DPTEMP -
      DPBESL = DPTEMP -
      DPP = P -
      DPX = X -
      DPCONS = (DPX/2.).P.DPP -
      DO THROUGH (LOOP),K=1,1,PROVIDED(.ABS.(DPTEMP*DPCONS).G.1..X
      .-6) -
      DPK = K -
      DPTEMP = -DPTEMP*DPX*(1./DPK)*(1./(4.*(DPP+DPK)))*DPX -
      DPBESL = DPBESL + DPTEMP -
LOOP   CONTINUE -
      DPBESL = DPBESL*DPCONS -
      BESL = DPBESL -
      NORMAL EXIT -
      END SUBPROGRAM -
C      -
      SUBROUTINE () = RESET.(ARGMAX) -
C      -
      DIMENSION (T(100)) -
      KE = ARGMAX/3.14159265 -
      T(2) = RFRS.(7.85398163) -
      DO THROUGH (END), K=2,1,PROVIDED (K.LE.KE) -
      SIGN = 1. -
      PROVIDED (K.E.K/2*2), SIGN = -1. -
      CONST = 1./(K*3.14159265+1.57079632).P.2 -
      S0 = 2. -
      S1 = 2./(2.*K+1.) -
      C = S1 -
      B1 = -.5 -
      SUM = S0 + S1*B1 -
      N = 1 -
TERM   N = N + 1 -
      TN = N -
      S2 = C.P.N-TN*(TN-1.)*CONST*S0 -
      B2 = -B1*(1.-.5/TN) -
      TEMP = B2 * S2 -
      SUM = SUM + TEMP -
      S0 = S1 -
      S1 = S2 -
      B1 = B2 -
      PROVIDED (.ABS.TEMP.G.1..X.-8), TRANSFER TO (TERM) -
      T(K+1) = T(K)+.31830989/SQRT.(2.*K+1.)*SUM*SIGN -
      CONTINUE -
      NORMAL EXIT -
C      -
      FUNCTION (C) = RFRESN.(ARG) -
C      -
      EXTERNAL (T) -
      PROVIDED (ARG.LE.7.85398163), TRANSFER TO (LESS) -
      T1 = ARG*.31830989-.5 -
      K = T1 -
      T2 = K -
      PC = (T1-T2)/(T2+.5) -
      A = (T1-T2)*3.14159265 -
      C2 = -COS.(A) -
      SIGN = 1. -
      PROVIDED (K.E.K/2*2), SIGN = -1. -
      C1 = .63661978/(2.*T2+1.) -
      B2 = -.5 -
      C3 = C1*SIN.(A) -
      C1 = C1*C1 -

```

```

S1 = 1.+C2 -
S2 = C3+C2*PC -
SUM = S1 + B2*S2 -
N = 1 -
TERM N = N + 1 -
TN = N -
B3 = -B2*(1.-.5/TN) -
S3 = PC.P.N*C2+TN*PC.P.(N-1)*C3-TN*(TN-1.)*C1*S1 -
TEMP = B3*S3 -
SUM = SUM + TEMP -
B2 = B3 -
S1 = S2 -
S2 = S3 -
PROVIDED (.ABS.TEMP.G.1..X.-6), TRANSFER TO (TERM) -
C = T(K) + .31830989/SQRT.(2.*T2+1.)*SUM*SIGN -
NORMAL EXIT -
LESS C = RFRS.(ARG) -
NORMAL EXIT -
END SUBPROGRAM -
C -
FUNCTION (C) = RFRS.(X) -
C -
H = SQRT.(2.*X/3.1415927) -
C = H -
DO THROUGH (LOOP), L=1,1..ABS.(H).G.1..X.-7 -
H=H*(((3.-4.*L)*X*X)/((4.*L+1.)*(2.*L)*(2.*L-1.))) -
C = C + H -
LOOP CONTINUE -
DONE NORMAL EXIT -
END SUBPROGRAM -
END SUBPROGRAM -
C -
SUBROUTINE () = UFRS.(ARGMAX) -
C -
DIMENSION (T(100)) -
KE = ARGMAX/3.14159265 -
T(2) = UFRS.(6.28318531) -
DO THROUGH (END), K=2,1.K.LE.KE -
SIGN = -1. -
PROVIDED (K.E.K/2*2), SIGN = 1. -
CONST = 1./(K*3.14159265).P.2 -
S0 = 2. -
S1 = 1./K -
C = S1 -
B1 = -.5 -
SUM = S0+S1*B1 -
N = 1 -
TERM N = N+1 -
TN = N -
S2 = C.P.N-TN*(TN-1.)*CONST*S0 -
B2 = -B1*(1.-.5/TN) -
TEMP = B2*S2 -
SUM = SUM+TEMP -
S0 = S1 -
S1 = S2 -
B1 = B2 -
PROVIDED (.ABS.TEMP.G.1..X.-8), TRANSFER TO (TERM) -
T(K+1) = T(K)+.31830989/SQRT.(2.0*K)*SUM*SIGN -
END CONTINUE -
NORMAL EXIT -
C -
FUNCTION (S) = UFRSN.(ARG) -
C -
EXTERNAL (T) -

```

```

PROVIDED (ARG.LE.6.2831854), TRANSFER (LESS) -
T1 = ARG*0.31830989 -
K = T1 -
T2 = K -
PC = (T1-T2)/T2 -
A = (T1-T2)*3.14159265 -
C2=-COS.(A)-
SIGN = -1. -
PROVIDED (K.E.K/2*2), SIGN = 1. -
C1 = 0.31830989/T2 -
B2 = -.5 -
C3 = C1*SIN.(A) -
C1 = C1*C1 -
S1 = 1.+C2 -
S2 = C3+C2*PC -
SUM = S1 + B2*S2 -
N = 1 -
TERM      N = N + 1 -
TN = N -
B3 = -B2*(1.-.5/TN) -
S3 = PC.P.N*C2+TN*PC.P.(N-1)*C3-TN*(TN-1.) *C1*S1 -
TEMP = B3*S3 -
SUM = SUM + TEMP -
B2 = B3 -
S1 = S2 -
S2 = S3 -
PROVIDED (.ABS.TEMP.G.1..X.-6), TRANSFER TO (TERM) -
S = T(<)+0.31830989/SQRT.(2.*T2)*SUM*SIGN -
NORMAL EXIT -
LESS      S = UFRS.(ARG) -
NORMAL EXIT -
END SUBPROGRAM -
C
FUNCTION (S) = UFRS.(ARG) -
C
H = SQRT.(2.*ARG/3.1415927)*ARG/3. -
S = H -
DO THROUGH (LOOP), L=1.1..ABS.(H).C.1..X.-7 -
H = H*(((1.-4.*L)*ARG*ARG)/((4.*L+3.)*(2.*L+1.)*(2.*L))) -
S = S + H -
LOOP      CONTINUE -
NORMAL EXIT -
END SUBPROGRAM -
END SUBPROGRAM -

```

APPENDIX II

This Appendix gives the Scatran computer subroutine for solving the set of three simultaneous complex equations given in Eq. (18). The subroutine was programmed by Richmond[8] and employs the Crout method for solving simultaneous linear equation[7]. The parameter IJK is the dimension vector of each equation and is usually equal to $N + 1$.

```

C J. RICHMOND CROUT METHOD PROGRAM *
C U(I) = SOLUTIONS, C = COEFS., N = NO. OF EQS., IJK = DIM. VECTOR -
C
C          C11 U1 + C12 U2 = C13
C          C21 U1 + C22 U2 = C23
C
SUBROUTINE(U,CX)=SSECM.(N,IJK)-
NN=N+1-
IJL=IJK+1-
COMPLEX(CX(0,IJK),C(462,IJL))-
DOTTHROUGH(SY),I=1,1,I.LE.NN-
DOTTHROUGH(SY),J=1,1,J.LE.NN+1-
SY C(I,J)=CX(I,J)-
    DOTTHROUGH(S118),L=1,1,L.LE.N-
    LLL=L-1-
    DO THROUGH(S118),I=L,1,I.LE.N-
    II=I+1-
    DO THROUGH(S117),K=1,1,K.LE.LLL-
    C(I,L)=C(I,L)-C(I,K)*C(K,L)-
S117 C(L,II)=C(L,II)-C(L,K)*C(K,II)-
S118 C(L,II)=C(L,II)/C(L,L)-
    DO THROUGH(S123),L=2,1,L.LE.N-
    I=NN-L-
    II=I+1-
    DO THROUGH(S122),K=II,1,K.LE.N-
S122 C(I,NN)=C(I,NN)-C(I,K)*C(K,NN)-
S123 CONTINUE-
    DOTTHROUGH(SX),LMN=1,1,LMN.LE.N-
SX U(LMN)=C(LMN,NN)-
    NORMALEXIT-
    ENDSUBPROGRAM-

```

REFERENCES

1. Rudduck, R. C., "Application of Wedge Diffraction to Antenna Theory," Report 1691-13, 30 June 1965, Antenna Laboratory, The Ohio State University Research Foundation; prepared under Grant Number NsG-448, National Aeronautics and Space Administration, Office of Grants and Research Contracts, Washington, D. C.
2. Ryan, C. E. Jr., and Rudduck, R. C., "Calculation of Radiation Pattern of a General Parallel Plate TEM Waveguide Aperture," Report 1394-11, 31 December 1963, Antenna Laboratory, The Ohio State University Research Foundation; prepared under Contract Number AF 33(657)-7829, Aeronautical Systems Division, Wright-Patterson Air Force Base, Ohio. (AD 133 716)
3. Ryan, C. E. Jr., and Rudduck, R. C., "Calculation of the Radiation Pattern of a General Parallel-Plate Waveguide Aperture for the TEM and TE_{01} Waveguide Modes," Report 1693-4, 10 September 1964, Antenna Laboratory, The Ohio State University Research Foundation; prepared under Contract Number N62269-2184, U. S. Naval Air Development Center, Johnsville, Pennsylvania.
4. Pauli, W., "On Asymptotic Series for Functions in the Theory of Diffraction of Light," Phys. Rev., 54, 1 December 1938, pp. 924-931.
5. Burke, J. E., and Keller, J. B., "Diffraction by a Thick Screen, A Step and Related Axially Symmetric Objects," EDL-E48, Contract DA36-039-sc-78281, Sylvania Electronic Systems, Mountain View, California. (March 1960).
6. Twersky, V., "Multiple Scattering of Waves and Optical Phenomena," J. Opt. Soc. Am., 52, No. 2, (February 1962) pp. 145-171.
7. Crout, P. D., "A Short Method for Evaluating Determinants and Solving Systems of Linear Equations with Real and Complex Coefficients," AIEE Trans., 60, (1941), pp. 1235-1241.
8. Richmond, J. H., "Scattering by an Arbitrary Array of Parallel Wires," Report 1522-7, 30 April 1964, Antenna Laboratory, The Ohio State University Research Foundation, prepared under Contract Number N123(953)-3166A, U. S. Navy Electronics Laboratory, Los Angeles 55, California. (AD 443 833).

9. Noble, B., Methods Based On the Wiener-Hopf Technique, Pergamon Press, New York, (1958), pp. 100-105.
10. Fleckner, O.L., Jr., "Computation of the Fresnel Integrals," M.Sc. Thesis, Department of Mathematics, The Ohio State University, Columbus, Ohio, (1965).